Context-free grammar

program	::=	**program id (** identifier_list **) ;** declarations subprogram_declarations compound_statment **.**
identifier_list	::=	**id** identifier_list_marked
identifier_list_marked	::=	**, id** identifiear_list_marked
	|	**ϵ**
declarations	::=	**var** identifier_list **:** type **;** declarations
	|	**ϵ**
type	::=	standard_type
	|	**array [num .. num] of** standard_type
standard_type	::=	**integer**
	|	**real**
subprogram_declarations	::=	subprogram_declaration **;** subprogram_declarations
	|	**ϵ**
subprogram_declaration	::=	subprogram_head declarations compound_statement
subprogram_head	::=	**function id** arguments **:** standard_type **;**
	|	**procedure id** arguments **;**
arguments	::=	**(** parameter_list **)**
	|	**ϵ**
parameter_list	::=	identifier_list **:** type parameter_list_marked
parameter_list_marked	::=	**;** identifier_list **:** type parameter_list_marked
	|	**ϵ**
compound_statment	::=	**begin** optional_statements **end**
optional_statements	::=	statement_list
	|	**ϵ**
statement_list	::=	statement statement_list_marked
statement_list_marked	::=	**;** statement statement_list_marked
	|	**ϵ**

statement	::=	id statement_marked
	|	compound_statement
	|	**if** expression **then** statement **else** statement
	|	**while** expression **do** statement
statement_marked	::=	[expression] **assignop** expression
	|	**assignop** expression
	|	**(** expression_list **)**
	|	**ϵ**
expression_list	::=	expression expression_list_marked
expression_list_marked	::=	**,** expression expression_list_marked
	|	**ϵ**
expression	::=	simple_expression expression_marked
expression_marked	::=	**relop** simple_expression
	|	**ϵ**
simple_expression	::=	term simple_expression_marked
	|	sign term simple_expression_marked
simple_expression_marked	::=	**addop** term simple_expression_marked
	|	**ϵ**
term	::=	factor term_marked
term_marked	::=	**mulop** factor term_marked
	|	**ϵ**
factor	::=	**id** factor_marked
	|	**num**
	|	**(** expression **)**
	|	**not** factor
factor_mark	::=	**(** expression_list **)**
	|	**[** expression **]**
	|	**ϵ**
sign	::=	**+**
	|	**-**